

Le tableur CALC d'OpenOffice (version 3.1.1)

Dans le cadre du Plan Académique de Formation, nous animons un stage « Créations d'activités intégrant les TIC » en Lycée Professionnel. Voici les documents que nous avons créés pour accompagner les stagiaires lors de ce stage.

- Les documents de *prise en main* proposent des activités de découverte du tableur complètement guidées. Ces documents peuvent être travaillés en présentiel ou à distance. Ils peuvent également être considérés comme des documents « ressources ».
- Les *applications* exploitent des applications concrètes avec une guidance plus relative.
- Quant aux *exercices*, un travail de recherche et de synthèse est demandé aux stagiaires.



Raphaël Boucly - raphael.boucly@wanadoo.fr
Philippe Dzewowski - philippe.dzewowski@orange.fr

Remarque : tous les documents, originaux et corrigés sont disponibles et modifiables.

Sommaire

Prise en main

saisie - recopie	page 1
adressage relatif - absolu	page 2
formule - fonction.....	page 3
l'alternative.....	page 4
fonction ESTTEXTE()	page 6
protection des cellules.....	page 7
représentation graphique	page 8
fonctions ALEA.ENTRE.BORNES et NB.SI.....	page 11
bouton et macro.....	page 13
compteurs et barres de défilement.....	page 16
formats de cellule	page 18
fonction DECALER() et INDEX()	page 20
fonction EQUIV().....	page 22
les matrices	page 23

1) Création d'un tableau :

	A	B	C	D	E
1	Nombre 1	14	140	280	
2	Nombre 2	7	0,7	0,14	
3	Somme				
4	Différence				
5	Produit				
6	Quotient				
7					
8					

A FAIRE !

AIDE

- a) Ouvrir OpenOffice.org Calc puis reproduire ce tableau.
- b) Compléter le par des formules dans la colonne B.

En **B3**, on saisira **=B1+B2**
De même, compléter les cellules **B4**, **B5** et **B6**.

- c) Recopier ces formules à droite.

Cliquer sur la poignée de recopie de la cellule (ici B3)

Remarque : On peut effectuer la même opération en sélectionnant plusieurs cellules.

- d) Renommer la feuille « Feuille1 » « exercice1 »

Clic-droit sur l'onglet « Feuille1 » puis Renommer la feuille

2) Tableau de proportionnalité

Ne tapez pas le signe €, choisissez plutôt le format monétaire en cliquant sur le bouton dans la barre d'outils « Formatage » !

	A	B		D	E
1	Masse des pommes en kg	1	4	5,5	17
2	Prix en euro	1,95 €			

A FAIRE !

AIDE

- a) Sélectionner la feuille « Feuille2 ». La renommer en « Exercice2 ».
- b) Reproduire ce tableau puis trouver la formule à mettre dans la cellule C2.
- c) Recopier ces formules à droite. Que remarquez-vous ?

Solution 1 : en C2, on saisira **=C1*1,95**
Inconvénient de cette méthode : imaginons que le prix du kilo passe à 2 €... Il n'y a pas de mise à jour automatique en C2...
Solution 2 : en C2, on saisira **=C1*B2**
Voir les inconvénients de cette méthode ci dessous

QUELQUES EXPLICATIONS

Cas de la solution 1 : les valeurs sont exactes mais si le prix change, il faut recommencer la saisie et la recopie des formules, c'est fastidieux !

Cas de la solution 2 : les valeurs sont fausses !!!


Pourquoi ?

En recopiant la formule **=C1*B2** vers la droite la formule, toutes les références aux colonnes (repérées ici par les lettres C et B) sont décalées !

Par exemple, en D2 on aura **=D1*C2** ce qui est faux !

Une solution ?

Elle consiste à figer (ou punaiser) la cellule B2 contenant le prix ! Pour cela on rajoutera des '\$' à l'adresse ! la formule en C2 sera donc **=C1*\$B\$2** . La référence **\$B\$2** ne changera pas lors de la recopie de la formule vers la droite. L'adressage **\$B\$2** est dit *adressage absolu* opposé à l'*adressage relatif* comme **B2** par ex.



Remarque :
l'appui sur les touches **Maj+F4** permettent d'afficher les '\$'...

3) Tableau de proportionnalité (suite)

	A	B	C	D	E
1	Masse des pommes en kg	2	4	5,5	17
2	Prix en euro	4,10 €			

A FAIRE !

AIDE

- a) Sélectionner la feuille « Feuille3 ». La renommer en « Exercice3 ».
- b) Reproduire ce tableau puis trouver la formule à mettre dans la cellule C2.
- c) Recopier cette formule à droite !



Vous pensez bien sûr au produit en croix, alors en vous aidant de l'exercice 2, vous devriez trouver la formule ! Attention aux '\$' !

Réponse : en C2, la formule est : =C1*\$B\$2/\$B\$1

4) Table de multiplication

- a) Il n'y a plus de feuilles disponibles. Insérons des feuilles. Pour en insérer une ou plusieurs, choisir le menu Insertion puis Feuille... Choisir alors le nombre de feuilles à créer puis valider.

- b) Voici le problème : il s'agit de compléter un tableau de tables de multiplication.

	A	B	C	D	E	F	G	H	I	J	K	L
1	x	0	1	2	3	4	5	6	7	8	9	10
2	0											
3	1											
4	2											
5	3											
6	4											
7	5											
8	6											
9	7											
10	8											
11	9											
12	10											
13												

Le problème est de trouver **une seule formule** que l'on recopiera dans tout le tableau... (2 solutions sont possibles)

- c) **Première solution :** on utilisera les références « mixtes » avec les '\$' !

Sur une feuille que l'on nommera « Exercice4 version1 », recopier le tableau ci-dessus.

A FAIRE !

AIDE

Utiliser l'adressage absolu avec les '\$' pour écrire la formule dans la cellule B2.

On a vu qu'avec un adressage absolu avec 2 '\$' (\$A\$1) : la cellule est complètement figée lorsque vous recopiez !

Par exemple, si en B4 on saisit la formule =**\$A1** seule la colonne A sera figée en cas de recopie !

Si par contre, en B4 on saisit la formule =**A\$1**, seule la ligne 1 sera figée en cas de recopie !

Observer le résultat ci-dessous :

Observer le résultat ci-dessous :

	A	B	C	D	E
1					
2					
3					
4		= \$A1	= \$A1	= \$A1	= \$A1
5		= \$A2			
6		= \$A3			
7		= \$A4			
8		= \$A5			
9		= \$A6			
10		= \$A7			
11		= \$A8			
12		= \$A9			
13		= \$A10			
14					

	A	B	C	D	E
1					
2					
3					
4		= A\$1	= B\$1	= C\$1	= D\$1
5		= A\$1			
6		= A\$1			
7		= A\$1			
8		= A\$1			
9		= A\$1			
10		= A\$1			
11		= A\$1			
12		= A\$1			
13		= A\$1			
14		= A\$1			

Essayer de trouver la « bonne formule » en B2 que l'on recopiera dans tout le tableau

Réponse : la formule en B2 sera donc : = **\$A2*B\$1**

d) *Deuxième solution* : on nommera les cellules !

A FAIRE !

AIDE

- 1) Sélectionner une nouvelle feuille.
La renommer en « *Exercice4 version2* ».
- 2) Recopier le tableau à nouveau le tableau du b).
- 3) On utilisera ici les « noms » de cellules.
Trouver la formule en B2 à recopier dans tout le tableau.
Remarque : La formule sera alors beaucoup plus simple et surtout plus compréhensible !

Sélectionner les cellules B1 à L1 (faites un cliquer-glisser pour « noircir » ces cellules)

Nommer la sélection « ligne »
Menu *Insertion* → *Noms* → *Définir*
ou le raccourci clavier *Ctrl+F3*
Taper « **ligne** » puis valider.

Nommer de la même façon les cellules A2 :A12 en **colonne**

Réponse : la formule en B2 sera donc : = ligne * colonne

5) Facture version 1

On vous propose de remplir dans un premier temps une facture simple.

	A	B	C
1			
2		Taux de TVA	19,6%
3			
4			
5	Prix HT	TVA	PTTC
6	120,00 €		
7	245,12 €		
8	58,00 €		
9	478,63 €		
10	35,90 €		
11	136,80 €		
12	402,00 €		
13	856,00 €		
14	79,60 €		
15		TOTAL	
16			


Taper 0,196 en C2
Appliquer le format pourcentage
(cliquer sur le bouton  de la barre d'outils 'Formatage')

On utilisera une fonction prédéfinie dans le tableur...
la fonction somme ou Σ

A FAIRE !

AIDE

- a) Sélectionner une nouvelle feuille.
La renommer en « *Facture version 1* ».
- b) Reproduire le tableau ci-dessus en respectant svp la disposition des cellules comme sur la copie d'écran. Ainsi les références proposées dans l'aide correspondront...
- c) Compléter le tableau !

- Pour calculer le montant de la TVA, on choisira :
 - d'utiliser les références absolues '\$C\$2' ... ou
 - de nommer la cellule C2 en 'tva' ...
Nous pouvons écrire en B6 : =A6*\$C\$2
- Pour calculer le PTTC, une simple somme suffira :
Nous pouvons écrire en C6 : =A6+B6
- Pour calculer le total en C15, nous utiliserons une fonction SOMME des cellules de C6 à C14.
 - Soit en cliquant sur le bouton Σ de la barre de formule 
 - Soit en tapant au clavier '=**SOMME(C6 :C14)**'

d) Une fois le tableau complété, changer le taux de TVA à 5,5% et observez les changements !

Attention à ce que l'option du menu *Outil* → *Contenu des cellules* → *Calcul automatique* soit bien coché ! ou gare...

6) Facture version 2

Cette fois, on a le choix entre deux taux de tva... Comme le montre le tableau si dessous.

	A	B	C	D
1				
2		Taux de TVA1	19,6%	
3		Taux de TVA2	5,50%	
4				
5	Prix HT	Code TVA	TVA	PTTC
6	120,00 €	1	23,52 €	143,52 €
7	245,12 €	2	13,48 €	258,60 €
8	58,00 €	1	11,37 €	69,37 €
9	478,63 €	1	93,81 €	572,44 €
10	35,90 €			
11	136,80 €			
12	402,00 €			
13	856,00 €			
14	79,60 €			
15			TOTAL	1 043,93 €
16				

Pour une meilleure lisibilité, on nommera ces 2 cellules tva1 et tva2.

On saisira 1 ou 2 dans cette colonne pour appliquer le taux de TVA 1 ou 2.

A FAIRE !


- a) Sélectionner une nouvelle feuille. La renommer en « Facture version 2 ».
- b) Reproduire le tableau ci-dessus en respectant svp la disposition des cellules comme sur la copie d'écran. Ainsi les références proposées dans l'aide correspondront...
- Remarque : vous pouvez faire des copier/coller de cellules de la feuille précédente pour gagner du temps !
- c) Compléter les cellules de B6 à B14 aléatoirement de 1 ou de 2 !
- d) Trouver la formule en C6... qui tient compte bien sûr du choix du taux de TVA. Formule qui sera recopiée jusqu'au bas du tableau.



Qu'est ce que j'en sais, moi ?

AIDE


- Nommer la cellule C2 tva1
Rappel : Menu Insertion → Noms → Définir ou le raccourci clavier Ctrl+F3
- Autre possibilité* : saisir au clavier tva1 à la place de C2 dans la barre d'outils 'barre de formule' et appuyer sur la touche Entrée.
- Nommer également C3 en tva2
- Pour trouver la formule en C6, il faut tester le contenu de la cellule B6 qui contient 1 ou 2. On utilisera une fonction logique importante, le SI. En voici sa syntaxe :



- SI(Test;Valeur_si_vrai; Valeur_si_faux)**
Test est une valeur ou expression pouvant renvoyer VRAI ou FAUX (autrement dit Test est un booléen).
Valeur_si_vrai (facultatif) est la valeur renvoyée si le test logique est VRAI.
Valeur_si_faux (facultatif) est la valeur renvoyée si le test logique est FAUX.
- En C6, la formule sera donc :
 '=SI(B6=1;A6*tva1;A6*tva2)'

QUELQUES EXPLICATIONS

Le *Si-Alors-Sinon* (appelée encore Alternative) peut être représenté par ce schéma. Cette représentation nous servira surtout par la suite



SI(B6=1;A6*tva1;A6*tva2)

- Remarquer que *le séparateur est le point virgule*
- Dans notre exemple, on ne teste que la valeur 1... cela suppose qu'il n'y ait que des 1 ou des 2 de saisis.
- Nous allons envisager les autres saisies possibles !


Dans notre exemple, on a B6=1 A6*tva1 A6*tva2

7) Améliorations de la facture version 2

Concernant la colonne TVA

- a) Envisageons maintenant la possibilité de mettre autre chose que des 1 ou 2 dans la colonne 'Code TVA'. Il va donc falloir modifier la formule **SI(B6=1;A6*tva1;A6*tva2)** qui ne traitait que les cas favorables. Il faut donc traiter AVANT d'exécuter cette alternative. On testera si la cellule B6 contient un 1 ou un 2... On utilise à cet effet ce que l'on appelle des **SI imbriqués**.

Mais un dessin vaut mieux qu'un long discours :



Il existe une fonction logique **OU** tableur : sa syntaxe est **OU(condition1;condition2;...)**.

La fonction OU renvoie :
« vrai » si l'une des conditions est réalisée
« faux » si aucune des conditions n'est réalisée.

Cela se traduit ici par **OU(B6=1;B6=2)**

Modifier alors la formule en C6 Compléter :


Voici la solution...

=SI(OU(B6=1;B6=2);SI(B6=1;A6*tva1;A6*tva2); "Code TVA invalide")

- b) Tout semble fonctionner maintenant correctement. Cependant, si on supprime le contenu de la cellule B6, on a le message « Code TVA invalide »... ce qui est gênant ! On voudrait ne rien afficher ! Nous allons donc avant tout tester si la cellule B6 contient quelque chose avant de continuer le traitement ! On utilisera une autre fonction : la fonction **ESTVIDE()** qui renvoie Vrai ou Faux suivant qu'il n'y ait rien ou pas dans la cellule mise entre parenthèses...

Attention : rien signifie que le contenu de la cellule a été supprimé par la touche **Supp** (ou **Del**) ! Evident ? peut être... mais si vous tapez un espace la cellule celle-ci n'est plus vide ... même si on ne voit rien !

- c) On imbrique donc un nouveau SI... on peut imbriquer jusqu'à 7 SI dans une formule ! Compléter le dessin suivant en dessinant ce nouveau Si... et essayer de trouver la nouvelle formule... qui se « rallonge » !



Solution : =SI (ESTVIDE(B6) ; " " ; SI(SI(OU(B6=1;B6=2);SI(B6=1;A6*tva1;A6*tva2); "Code TVA invalide")))

Concernant la colonne TVA

La formule en D6 (PTTC) est '=A6+C6' (PHT+TVA), ce qui est logique mais si on ne saisit pas un code de TVA valable, l'affichage de la TVA a été résolu précédemment **mais** le PTTC affiche quand même le PHT... Pourquoi ? Parce que le tableur considère dans une opération comme valant 0 toute cellule contenant du texte ou une formule !

Quand faut-il donc afficher en D6, le résultat du PHT + TVA ?



Indication : une solution consiste à utiliser la fonction **ESTTEXTE()**.
Que fait cette fonction ?

- Pour cela, utiliser l'assistant de fonction
- Sélectionner la fonction dans la liste déroulante
Une explication succincte est proposée



- En cliquant sur suivant, vous accédez à un assistant d'écriture...

Réponse : La formule en D6 sera donc : =SI(ESTTEXTE(C6) ; " " ; A6 + C6)

Infos sur les fonctions...


Le tableur possède plus de **360 fonctions prédéfinies** !... Vous pouvez également définir vos propres fonctions ! Pensez à utiliser l'assistant de fonctions ou en appuyant sur la **touche F1** pour avoir une aide beaucoup plus complète sur les fonctionnalités du tableur !

Aide à la saisie du code de la TVA...

Pour éviter une erreur de saisie, on peut créer une cellule avec **liste déroulante** dans laquelle sont affichées les valeurs possibles...

1				
2		Taux de TVA1	19,6%	
3		Taux de TVA2	5,50%	
4				
5	Prix HT	Code TVA	TVA	PTTC
6	100,00 €			
7	245,12 €			
8	58,00 €			

- Pour cela, sélectionner les cellules B6 à B14.
- Aller dans le menu **Données** → **Validité**.
- Sélectionner '**Liste**' puis entrer les valeurs de la liste



- Cocher, décocher 'Affichage de la liste de sélection'. Tester
- Cliquer sur les autres onglets. Ils permettent d'afficher des messages d'aide ; Essayer et tester !

Protection de cellules...

- Il est souvent intéressant d'empêcher de modifier certaines cellules... celles qui contiennent des données importantes, des formules par exemple. L'utilisateur ne pourra modifier que les cellules « autorisées ».
- Par défaut, toutes les cellules d'une feuille d'un classeur sont « verrouillées »... cela signifie que si vous « protégez » votre feuille de, l'utilisateur ne pourra pas modifier son contenu !

A FAIRE !

AIDE


a) On désire protéger les cellules contenant les calculs de la TVA, du PTTC et du TOTAL. Seules les cellules du Prix HT et du code TVA pourront être modifiées !

Consulter l'aide.....

b) Tester l'option « Masquer les cellules » qui peut parfois s'avérer fort utile (rappel : n'est valable qu'après protection de la feuille)

Il faut pour cela faire dans l'ordre :

- « déverrouiller » toutes les cellules que l'on veut pouvoir modifier. !
Sélectionner les cellules de A6 à B14 (en cliquant-glissant sur la zone A6 :B14) puis aller dans le menu **Format** → **cellules** et cliquer sur l'onglet « **Protection** »



Décocher l'option « Verrouiller »

- **Protéger** la feuille (avec un mot de passe éventuel. (menu **Outils** → **Protéger le document** → **Feuille** avec ou sans mot de passe !)

EXERCICE

Une fois l'exercice réalisé, les taux de TVA en C2 et en C3 sont également « bloqués » !
Faites en sorte que l'on puisse modifier ces taux...


8) Représentations graphiques : les courbes.

Nous allons aborder les représentations graphiques.

Un tableur ne sait représenter graphiquement que des tableaux de valeurs. Par exemple, le tableur ne peut pas représenter une fonction définie par $f(x) = -3x^2 + 7x - 9$: il faut remplir d'abord un tableau de coordonnées !

On vous propose de créer le graphique ci-dessous à partir de ce tableau.

	A	B	C	D	E	F	G	H
1	Côté du carré (cm)	0,5	1	2	2,5	3	5	6
2	Périmètre du carré (cm)							
3								



Remarquer que les valeurs du côté ne sont pas incrémentées de façon régulière...

A FAIRE !

AIDE


- Sélectionner une nouvelle feuille. La renommer en « Graphique ».
- Reproduire le tableau ci-dessus. Compléter la ligne de calcul du périmètre.
- Créer le graphique en utilisant l'assistant graphique.

Sélectionner le tableau (des cellules A1 à H2) puis

- cliquer sur le bouton  de la barre d'outils
- ou menu **Insertion** → **Diagramme...**

Voici les différentes étapes de création d'un graphique :


• Étape 1



Choisir la catégorie type « Diagramme XY »

Choisir une sous catégorie de graphique


• Étape 2



Bien préciser que les données sont en ligne.


La première colonne ne contient pas de données

• Étape 3




Cocher éventuellement pour afficher les lignes de quadrillage ...

• Étape 4



Pour réduire la taille des carrés représentant les points du graphique :

Double clic sur le graphique, (le bord du graphique devient grisé), puis double clic sur un carré, sélectionner l'onglet lignes. Mettre les largeurs et hauteurs des symboles à 0,1 par exemple.



REMARQUES


Vous pouvez modifier toutes ces options une fois le graphique terminé !

- **Sélectionner le graphique** en double-cliquant dessus (ou **clic droit** sur le graphique puis **Editer**). Le graphique s'entoure d'une bordure grise.
- Cliquer alors sur les différents éléments du graphique pour les sélectionner (ou appuyer successivement sur la touche **Tabulation** ).
-
- *En exercice*, changez la couleur de fond, faites commencer la graduation de l'axe des ordonnées à partir de 0...

Ajout d'une courbe...

On désire rajouter une ligne au tableau ainsi que sa courbe dans le graphique comme suit :

	A	B	C	D	E	F	G	H
1	Côté du carré (cm)	0,5	1	2	2,5	3	5	6
2	Périmètre du carré (cm)	2	4	8	10	12	20	24
3	Surface du carré (cm ²)	0,25	1	4	6,25	9	25	36



A FAIRE !

- a) Ajouter la ligne 'Surface du carré (cm²)'. Remplir la cellule B3. Recopier la formule dans tout le tableau.
- b) Rajouter cette ligne dans la représentation graphique.

AIDE


- En B3, on pourra taper « =B1^2 ». Le symbole puissance est obtenu en appuyant sur la touche **AltGr** + $\frac{9}{5}^{\wedge}$.
- On peut aussi utiliser la fonction **PUISSANCE** (voir l'aide dans l'assistant de fonctions)

Rajout de la courbe :

- Sélectionner les cellules **A3:H3**.
- **Cliquer** sur la sélection et **glisser** au dessus du graphique, **appuyer** sur la touche **Ctrl** avant de **relâcher la souris !** Le graphique est rajouté !
- On peut aussi sélectionner tout le tableau (**A1:H3**) et pratiquer un cliquer-glisser sur le graphique !

9) Création d'un diagramme en bâtons

On utilise ici le tableur pour simuler un jet de dé. On jette ici un dé successivement 75 fois et on relève dans un tableau le nombre d'apparitions de chaque face. Un diagramme en bâtons complète l'étude.



A FAIRE !

a) Ouvrir OpenOffice.org Calc puis reproduire ce tableau.

Enregistrer le fichier sous le nom : Lancer d'un dé.

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O
1	5	1	1	1	3	5	3	4	1	1	3	1	4	4	5
2	4	1	4	2	3	5	6	5	2	3	2	1	1	3	5
3	6	5	6	5	4	5	6	4	6	6	5	4	1	2	2
4	5	2	1	4	1	4	1	5	5	5	1	3	1	5	3
5	1	3	1	2	3	2	1	3	6	6	4	4	3	1	4

b) Calculer le nombre de lancers.


Nombre de tirages	75
-------------------	----

c) Reproduire le tableau ci-dessous.

	P	Q	R
8	Chiffres Sortis	Nombre de fois	Fréquence %
9	1	13	17%
10	2	10	13%
11	3	19	25%
12	4	11	15%
13	5	13	17%
14	6	9	12%
15	TOTAL	75	100,00%

d) Sélectionner la colonne des fréquences cellules puis créer le graphique ci-dessous.

Chiffres Sortis	Nombre de fois	Fréquence %
1	13	17%
2	10	13%
3	19	25%
4	11	15%
5	13	17%
6	9	12%
TOTAL	75	100,00%



AIDE

- Pour obtenir un nombre aléatoire compris entre 1 et 6, entrer dans la cellule A1 : **=ALEA.ENTRE.BORNES(1;6)**
- Recopier cette formule sur la plage **A1:O5**.
- Pour déterminer le nombre de valeurs dans le tableau, entrer dans la cellule R2 : **=NBVAL(A1:O5)**
- Pour calculer le nombre de sortie du chiffre 1, entrer dans la cellule Q9 : **=NB.SI(\$A\$1:\$O\$5 ;P9)**
- Pour calculer le nombre de sortie des autres chiffres, recopier la formule précédente jusque la cellule Q14.
- Pour calculer la somme des sorties des chiffres, entrer dans la cellule Q15 : **=SOMME(Q8 :Q14)**

- Utiliser le format pourcentage pour le calcul des fréquences.
- En utilisant l'assistant graphique , créer le diagramme en bâtons.
- Modifier les graduations de l'axe des ordonnées comme indiqué ci-dessous.

Axe Y

Échelle

Minimum : 0% Automatique

Maximum : 30% Automatique

Intervalle principal : 10% Automatique

Nombre d'intervalles secondaires : 5 Automatique

Décocher les coches « Automatique » et mettez les valeurs indiquées.

IMPORTANT

Pour effectuer une nouvelle simulation, appuyer sur les touches **Ctrl + Maj + F9**. Toutes les formules de la feuille sont recalculées !

Les lancers de dés, le tableau récapitulatif ainsi que le graphique sont mis à jour !


En exercice, rajouter 2 lignes de « lancers de dés » (c'est-à-dire qu'il y aura maintenant 105 lancers de dés simulés). *Modifier les formules en conséquence...*

Remarque générale sur les tableurs-grapheurs :

Un tableur (quel qu'il soit) ne sait pas tracer de « vrais » histogrammes mathématiques ! Pour en tracer, il faut modifier certains paramètres du graphique. En fait, il faut diminuer l'espacement entre les différents rectangles !

10) Quelques améliorations possibles

On se propose de modifier le graphique précédent pour obtenir cet histogramme.



Les rectangles sont **contigus** et de **couleurs différentes** !

Modifier l'espacement des rectangles.

- Sélectionner le graphique (*double-clic*)
Le graphique s'entoure d'une *bordure grise*.
 - Sélectionner l'ensemble des rectangles (appui successifs sur la touche **Tabulation**  ou double-clic sur l'un des rectangles) comme suit :
- | Category | Percentage |
|----------|------------|
| Face 1 | ~11% |
| Face 2 | ~14% |
| Face 3 | ~17% |
| Face 4 | ~19% |
| Face 5 | ~21% |
| Face 6 | ~17% |
- Faire alors un *clic droit* sur l'un des rectangles puis sélectionner « **Propriétés de l'objet** »
 - Dans l'onglet « **Options** », fixer à **0%** la valeur dans la rubrique « **espacement** ».

Modifier la couleur des rectangles.

- Après avoir sélectionné l'ensemble des rectangles, double-cliquer sur un des rectangles...
- Activer l'onglet « **remplissage** » et changer de couleur....

11) Boutons et macros

On se propose de rajouter un bouton à l'exercice sur les lancers d'un dé. Ce bouton simule un nouveau tirage.

Ce travail se fait en 2 étapes :

- Créer une macro qui simulera les nouveaux lancers du dé.
- Affecter à un bouton cette macro.

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q	R
1	4	6	4	1	2	1	6	1	5	4	1	4	1	5				
2	6	2	6	4	3	4	6	2	6	3	3	1	6	3	4		Nombre de tirages	105
3	1	6	4	2	1	2	6	5	2	1	3	3	1	5	3			
4	4	2	3	2	3	5	3	3	5	3	4	3	1	2	3			
5	4	1	5	4	2	5	1	5	1	3	4	3	2	1	3			
6	3	6	1	4	2	6	3	4	3	5	4	5	1	5	3			
7	3	2	6	4	3	5	5	2	3	5	2	1	4	5	4			
8																Chiffres sortis	Nombre de fois	Fréquence %
9																Face 1	18	17%
10																Face 2	15	14%
11																Face 3	24	23%
12																Face 4	19	18%
13																Face 5	17	16%

A FAIRE !

a) Ouvrir le fichier : Lancer d'un dé.ods précédemment sauvegardé.

b) Créer une macro 'tirage_au_sort' .

Le nom de la macro ne peut pas contenir d'espaces ! Utiliser le **tiret bas** !


AIDE

Utilisons l'enregistreur automatique de macro.

- Menu **Outils** → **Macros** → **Enregistrer une macro**. Une barre d'outils s'affiche qui vous permettra d'arrêter l'enregistrement de la macro.




- Toutes les actions que vous faites (saisie au clavier, déplacements dans les menus, ...) sont alors enregistrées automatiquement jusqu'à ce que l'on clique sur le bouton '**Terminer l'enregistrement**' !
- La seule commande ici à enregistrer est de taper au clavier **Ctrl+Maj+F9**
- Arrêter l'enregistrement.
- Une nouvelle fenêtre s'ouvre :



Dans la zone Nom de la macro, supprimer le nom Main par le nom : tirage_au_sort, puis cliquer sur Enregistrer

Pour tester la macro :

Outils → Macros → Exécuter la macro




Sélectionner la macro tirage_au_sort, puis cliquer sur Exécuter.

Pour pouvoir utiliser la macro sur un autre ordinateur, il faut lier la macro au fichier « Lancer d'un dé».

Enregistrer la macro :

- Menu **Outils** → **Macros** → **Enregistrer une macro**.
- Puis Ctrl+Maj+F9
- Puis appuyer sur Terminer l'enregistrement.
-

Une fenêtre s'affiche :



Cliquer sur le bouton : Nouveau module



Cliquer sur OK.

Puis remplacer le nom Main par le nom de la macro : tirage_au_sort puis enregistrer.

La macro sera enregistrée avec le fichier 'Lancer d'un dé' donc elle sera opérationnelle si on copie le fichier sur un autre ordinateur !

REMARQUE

Si vous voulez voir les lignes de programmes générées par l'enregistreur, allez dans le menu **Outils**→**Macros**→**Gérer les macros**→**OpenOffice.org Basic**. Sélectionner la macro et cliquer sur le bouton 'Editer' !

Le langage de programmation s'appelle *langage de script OOoBasic* !!

Bonne lecture... et surtout bon courage !

A FAIRE !

Création du bouton

c) Afficher la barre d'outils 'Contrôles de formulaire'

d) Dessiner le bouton 'Nouveaux lancers'

e) Affecter la macro '**Tirage_au_sort**' à ce bouton.


AIDE

- Menu **Affichage**→**Barres d'outils**→ **Contrôles de formulaire**




Le bouton de '**mode de conception**' permet soit de construire un contrôle, soit de le tester !

- Cliquer sur le bouton et en dessiner un sur la feuille !



- Le bouton étant *sélectionné* : clic droit puis choisir l'option **contrôle**. Dans l'onglet 'Général', changer le titre du bouton !




- Dans l'onglet '**Evènements**',
Fac à l'évènement 'Bouton de souris relâché', cliquer sur le bouton



- Cliquer sur le bouton '**Macro**'



- Sélectionner alors la **macro concernée**...dans la bonne **bibliothèque**



- Cliquer sur le bouton '**OK**' !
- Fermer la fenêtre.

f) Tester le bouton.

- Pour pouvoir tester le bouton, cliquer sur le bouton 'mode de conception'  qui est en fait une bascule.

12) Créations de compteurs et de barres de défilement

	A	B	C	D	E	F	G	H
1		f(x)=	-7	x	+2			
2								
3								
4								
5								
6								
7					La fonction f est une fonction :	affine		Très bien
8								

A FAIRE !

- a) Ecrire dans les cellules B1 à E1, l'expression d'une fonction affine.
- b) Créer un compteur permettant de faire varier le coefficient directeur de la fonction affine entre -10 et 10 avec un pas de 1.
- c) Créer un compteur permettant de faire varier le coefficient de l'ordonnée à l'origine de la fonction affine entre -10 et 10 avec un pas de 1.

AIDE

Affichage → Barre d'outils → Contrôle de formulaire

Cliquer sur le compteur : 

Dessiner le compteur verticalement puis faire un clic droit sur le compteur,

Contrôles → Onglet Général

Valeur min.....	-10	
Valeur max.....	10	

Onglet Données → Cellule liée : C1

Fermer la fenêtre en cliquant sur la croix.

Désactiver le mode Conception  (c'est une bascule)

Le compteur est maintenant opérationnel.

Activer le mode Conception.

Faire un copier/coller du premier bouton pour avoir les mêmes dimensions que le premier bouton.

Puis faire les modifications pour faire varier la cellule E1 entre -10 et 10.

Il y a un problème d'affichage ! En effet, dans la cellule E1, le signe + d'un nombre positif ne s'affiche pas. En revanche, il n'y pas de problème pour les nombres négatifs. Il faudrait forcer l'affichage du signe + pour les nombres positifs.

Pour y remédier :

Dans la cellule E1, clic droit → Formater les cellules

Onglet Nombres :

Description de format
"+"Standard;"-"Standard

Sélectionner les cellules B7 à F7, puis les fusionner en cliquant sur le bouton : 

d) Fusionner les cellules B7 à F7.

e) Ecrire en B7 : « La fonction f est : »

f) Créer une zone de liste avec les termes :
linéaire, affine, constante

Affichage → Barre d'outils → Contrôle de formulaire



Désactiver l'assistant (si nécessaire) : cliquer sur



Cliquer sur la zone de liste :



Dessiner la zone de liste avec la souris au dessus de la cellule G7.

Faire un clic droit sur la zone de liste, puis

Contrôles → Onglet général

Ecrire : linéaire, affine, constante dans la zone Entrée de Liste.



Astuce : pour écrire les termes les uns en dessous des autres, faire Maj+Entrée.

Fermer la fenêtre.

Désactiver le mode conception.



Prolongement : Faire varier les compteurs entre -10 et 10 avec un pas de 0,5.

Autre exercice dans une autre feuille de calcul :

Créer l'affichage d'une fonction du second degré avec les coefficients variant entre -10 et 10 avec un pas de 1. A la place des compteurs, pour changer, nous créerons des barres de défilement.

A	B	C	D	E	F	G	H
	f(x)=	-6	x ²	+ 10	x	+ 7	
		▲		▲		▲	
		■		■		■	
		▼		▼		▼	

13) Les formats prédéfinis :

- Nous avons vu jusqu'à présent le *format monétaire* qui permettait d'afficher un nombre avec deux décimales suivi du signe « € » ! Les valeurs négatives sont d'ailleurs affichées en rouge. L'intérêt d'un tel format est double :
 - Il permet une meilleure lecture des données dans un tableau.
 - Les cellules concernées sont toujours considérées comme des nombres, donc on peut calculer en leur faisant référence !

Exemple :

La description de ce format est accessible par le menu *Format* → *Cellules* dans l'onglet 'Nombres' :



Remarquer la suite de # et de 0 !

- Un 0 affiche les zéros supplémentaires si un nombre possède moins de chiffres que de zéros définis dans le format
- Un # n'affiche aucun zéro supplémentaire

Exemples (extraits de l'aide d'Open Office)

Format numérique	Code de format
3 456,8 pour 3456,78	# ####,##
9,900 pour 9,9	#,000
13,0 pour 13 et 1234,57 pour 1234,567	#,0#
5 3/4 pour 5,75 et 6 3/10 pour 6,3	# ???/???
0,5 pour ,5	0,###

14) Les formats personnalisés :

Nous pouvons rajouter des formats personnalisés... Pourquoi faire ? Par exemple :

- en calculs numériques : afficher un résultat avec une précision donnée : 1,2,... chiffres après la virgule...
- en sciences physiques : afficher par exemple '3 V', '0,2 A', '250 mA', '100 Ω', '150 daN', '25 m/s', 'kg'...
- en statistiques : afficher par exemple les crochets et le point virgule de la classe [10 ; 20] en saisissant seulement 10 et 20....

En exemple, créons ce tableau dans lequel

- on ne saisit que la valeur 12 en B3, 33 en C3. Les unités s'affichent automatiquement !
- la valeur de I est calculée automatiquement et est affichée avec 3 chiffres après la virgule...
-

	A	B	C	D
1				
2	I=U/R	U	R	I
3		12 V	33 Ω	0,364 A
4				

A FAIRE !

a) Reproduire ce tableau dans une nouvelle feuille nommée 'Format personnalisé'

b) Afficher automatiquement le V de la tension U.


c) Afficher automatiquement le Ω de la tension U.

d) Forcer l'affichage de l'intensité avec 3 chiffres.

AIDE

- Sélectionner la cellule B3, puis dans le menu **Format**→**Cellules** (ou un clic droit puis un clic sur 'Formater les cellules').

- Dans l'onglet « nombres », modifiez la « description de format »



A la suite de 'Standard' rajouter " V" (ne pas oublier l'espace). Valider. Tester

- Pour le signe Ω , on mettra une police d'affichage spéciale dans la cellule C3 : la police Symbol ! Le 'W' (w majuscule est Ω dans cette police !).
- Refaire comme pour la tension, la description sera alors **Standard " W"**
- Pour l'intensité, le " A" ne pose pas de problème... Par contre, pour forcer l'affichage à 3 décimales il faut remplacer 'Standard' par 0,000.
Le 0 force l'affichage d'un chiffre ! Par exemple, 1,2 sera affiché 1,200 et 1/3 en 0,333...
Le format en D3 sera donc

**REMARQUES**

Sur internet, vous trouverez des sites d'aide sur ce sujet et sur le tableur en général... un site intéressant est celui-ci
<http://christianwtd.free.fr/index.php>

ou encore le forum des utilisateurs de Calc :

<http://user.services.openoffice.org/fr/forum/viewforum.php?f=4>

En exercice essayer :

- de faire afficher les grands nombres par paquets de 3 chiffres... enfin, normalement !
- d'afficher un numéro de téléphone en séparant chaque paire de chiffre...

Enfin, plusieurs exercices vous sont proposés dans le dossier suivant. Nous vous conseillons de les faire en vous aidant bien sûr de cette prise en main... et de nous !

Bon courage...

raphael.boucly@wanadoo.fr
philippe.dzewowski@orange.fr

14) La fonction *DECALER* :

Nous vous proposons de vous faire découvrir la fonction **DECALER** assez déroutante au départ mais très puissante.

Son rôle : La fonction décaler permet de définir une plage en fonction d'une référence de départ, d'un décalage de ligne et de colonne par rapport à cette référence, ainsi que d'une définition de la hauteur et de la largeur de cette plage. Si la hauteur et la largeur ne sont pas définies, il y aura un simple décalage de la cellule.

Pour comprendre son fonctionnement, consulter le fichier *decaler.ods* dans le dossier « **Stage Tableur** » de votre **H:\Travail** ! Changer les valeurs dans les cellules jaunes pour comprendre le rôle de cette fonction.

8									
9	Ex : =DECALER(B20;ligne;Colonne;Hauteur;Largeur)								
10	Référence renvoyée : \$C\$21:\$J\$23								
11									
12	Référence de départ : B20								
13	Lignes : 1								
14	Colonnes : 1								
15	Hauteur : 3								
16	Largeur : 8								
17									
18									
19									
20	Janvier	Février	Mars	Avril	Mai	Juin	Juillet	Août	
20	Var 1	600	600	900	1100	200	800	300	300
21	Var 2	900	700	500	800	300	600	700	400
22	Var 3	1100	300	1000	800	700	900	1000	900
23	Var 4	300	700	300	800	300	400	300	1000
24	Var 5	1100	300	1100	500	1100	1200	1100	500

Modifier ces valeurs et visualiser sur le tableau le changement obtenu.

Cette fonction prend tout son sens lorsqu'elle est combinée avec une autre fonction. Par exemple, la fonction somme Σ.

28	Exemple avec la fonction somme :
29	=SOMME(DECALER(B20;B13;B14;B15;B16))
30	13 400

Changer la fonction *Somme* en fonction *Moyenne* par exemple....

15) La fonction *INDEX*

Son rôle : La fonction *INDEX* a pour but de renvoyer le contenu d'une cellule à l'aide de ses coordonnées dans une plage.


Consulter le fichier *indexequiv.ods* dans le dossier « **Stage Tableur** » et tester cette fonction en changeant les valeurs des cellules jaunes.

3	Modifiez la valeur des cellules jaunes pour comprendre les impacts.					
4						
5						
6	Ligne :	5				
7	Colonne :	3				
8						
9	Résultat :	527 =INDEX(D13:G22;B6;B7)				
10						
11						
12						
13	Lignes	1	678	493	58	981
14		2	172	41	191	240
15		3	464	661	97	716
16		4	776	80	747	39
17		5	731	693	527	873
18		6	337	465	797	27
19		7	136	165	196	595
20		8	188	420	671	177
21		9	685	808	986	592
22		10	109	221	624	380

Cette fonction permet donc de lire le contenu d'une cellule correspondant à l'intersection de la ligne et de la colonne dans la **matrice** indiquée (ici D13 :G22) !

16) Exemple d'utilisation de la fonction INDEX : un graphique interactif !

A partir d'un tableau de notes d'élèves, nous allons afficher l'histogramme en fonction du nom de l'élève.

**Comment faire ?**

Etape 1 : Créer le tableau avec les noms et notes

Etape 2 : En A7, taper « choix de l'élève » et entrer un numéro qui correspondra au numéro de l'élève.

Etape 3 : En A10, on utilisera la fonction **INDEX (Tableau ; No_Ligne ; No_Colonne)** qui permet de retrouver la valeur d'une cellule dont les coordonnées sont indiquées par le **No_Ligne** et le **No_Colonne**. Dans l'exemple, le **No de colonne** n'est pas nécessaire, puisque le tableau n'a qu'une colonne.

La formule en A10 est donc « =INDEX(A2 :A5 ;\$B\$7) ». Comme on a figé la cellule B7 avec les \$, il suffit de recopier cette formule vers la droite.

TESTER en changeant le numéro de l'élève en B7 !

Etape 4 : En A12, faites afficher de façon interactive le nom de l'élève après le texte *Résultats de l'élève...*

Etape 5 : Créer le graphique comme ci-dessus. Attention aux sélections disjointes (rappel : utiliser la touche Ctrl).
Afficher la note au dessus de chaque rectangle. Fixer l'échelle entre 0 et 20...

Etape 6 : Créer la zone de liste.

Afficher la barre d'outils « Contrôles de formulaire »

Dessinez une zone de liste

Définissez ses propriétés comme suit :



Et voilà, tester : c'est magique !

17) La fonction EQUIV

L'onglet suivant vous donne un aperçu du rôle de cette fonction.

	A	B	C	D	E	F	G	H
1	Illustration de la fonction EQUIV : Equiv(valeur_cherchée;tableau_recherche;type)							
2								
3	1. Valeur exacte							
4								
5	La fonction Equiv renvoie la position de la valeur cherchée dans la matrice.							
6								
7	Position :	1	2	3	4	5		
8		5	7	15	12	186		
9								
10	Choisissez la valeur que vous chercher...							
11		Valeur cherchée		12				
12		Position dans la matrice :		4	=EQUIV(D11;B8:F8;0)			
13		Type :		0				
14								
15	2. Valeur approchée							
16								
17	La valeur "type" dans la formule, permet de renvoyer une valeur exacte (cf. ci-dessus) ou une valeur approchée.							
18	Fonctionnement : la recherche s'effectue jusqu'à ce que la valeur							
19	cherchée soit plus grande que la valeur étudiée.							
20								
21		1	2	3	4	5		
22		5	7	15	12	16		
23								
24								
25		Valeur cherchée		12				
26		Position dans la matrice :		2	=EQUIV(D25;B22:F22;1)			
27		Type :		1				
28	Ex : Si la valeur cherchée est 12, la formule renvoie 2, car 15 est le premier chiffre de la matrice							
29	plus grand que 12. La fonction s'arrête donc à ce "plafond".							
30								

Compléments sur l'argument type :

Type : valeurs 1, 0 ou -1 :

- 1 ou absent : on suppose la première colonne de la matrice de recherche triée par ordre croissant. Dernière valeur inférieure ou égale au critère de recherche est retournée.
- -1 : on suppose la colonne triée par ordre décroissant. (mode Microsoft Excel). Première valeur supérieure ou égale est renvoyée
- 0 : une concordance exacte doit être trouvée ; si le critère de recherche est détecté à plusieurs reprises, la fonction renvoie la première occurrence. Caractères génériques autorisés.

ATTENTION DONC A CE QUE LES DONNÉES NUMÉRIQUES SOIENT TRIÉES !

18) Lecture d'un tableau à double entrée.

Les fonctions INDEX et EQUIV sont souvent utilisées ensemble... Voyons cela dans un exemple...

	A	B	C	D	E	F	G
1		Français	Mathématiques	Sciences Phys	Anglais	Technologie	CMC
2	Pierre	12	8	4	12	15	9
3	Paul	8	9	12	11	17	9
4	Jacques	11	13	11	10	16	10
5	Louis	10	11	15	9	19	12
6							
7							
8	Nom	Matière	Note				
9	Paul	CMC	9	=INDEX(notes;EQUIV(A9;noms;0);EQUIV(B9;matières;0))			

A vous de recréer ce tableau (un copier/coller sera le bien venu). Les noms et matières seront choisies dans des listes pour éviter les fautes de frappe... Bon courage !

Nom	Matière
Paul	CMC
Pierre	
Paul	
Jacques	
Louis	

19) Les matrices

Le calcul matriciel peut nous aider par exemple à résoudre un système d'équations linéaires !

Il existe des fonctions spécifiques pour les matrices.

Par exemple :

- **PRODUITMAT(matrice1 ;matrice2)** effectuera le produit des matrice1 et 2. On rappelle que le nombre de colonnes de la matrice1 doit correspondre au nombre de lignes de la matrice2 !
- **INVERSEMAT(matrice)** calcule la matrice inverse d'une matrice !
- **DETERMAT(matrice)** renvoie le déterminant d'une matrice *carrée* !

Comment saisir des formules matricielles ?

On se place dans une cellule et :

- si on saisit manuellement la formule, il faut valider celle-ci par la combinaison de touches **Ctrl+Maj+Entrée** ! Sinon, seule la cellule active sera calculée !
- si on utilise l'assistant de fonction, il ne faut pas oublier de cocher la case matrice pour effectuer le calcul matriciel !

Dans tous les cas des **accolades** doivent s'afficher dans la barre de formules !

Exemple, 

Illustrons cela sur un exemple. Soit à résoudre le système
$$\begin{cases} 2x - 3y = 1 \\ -3x + 4y = -2 \end{cases}$$

Si on appelle **A** la matrice $\begin{pmatrix} 2 & -3 \\ -3 & 4 \end{pmatrix}$, **B** la matrice $\begin{pmatrix} 1 \\ -2 \end{pmatrix}$ et **X** la matrice $\begin{pmatrix} x \\ y \end{pmatrix}$; le système peut s'écrire

A . X = B. La solution peut alors s'écrire **X = A⁻¹ . B**

Saisissez dans une feuille la matrice A et B et essayer de retrouver les solutions (ici x=2 et y=1)

	A	B	C	D	E	F	G	H	I	J	K	L	M	N
1								matrice A		matrice B				
2		2x	-3y	=	1			2	-3		1			
3		-3x	+4y	=	-2			-3	4		-2			
4														
5														
6														
7								A ⁻¹		B	Produit de A ⁻¹ B			
8								4	-3	1	x=2			
9								-3	-2	-2	y=1			
10														
11														
12								A			A ⁻¹		Mat. Identité	
13								4	-3	X	2	-3	=	1 0
14								-3	-2		-3	4		0 1
15														
16														
17														
18														